
Related Documents 

Through related documents, users can quickly understand the positioning of this manual and the 
correlation between manuals and products. Go to HIWIN MIKROSYSTEM’s official website → Download 
→ Manual Overview for details (https://www.hiwinmikro.tw/Downloads/ManualOverview_EN.htm). 

Preface 

This manual explains the operation of software SPiiPlus MMI Application Studio when E series EtherCAT 
drive is used with ACS controller. The contents in this manual, including project creation of controller, 
communication setup, parameters setup, test run, are arranged in accordance with the procedure of 
complete machine setup. For detailed information on E series servo drive, please refer to the related user 
manuals. 

Specifications of Software/Hardware 

Name Version of Software/Firmware 

E1 Series EtherCAT Drive 

Software (Thunder): 1.13.7.0 or above 

Firmware: 2.13.6 or above 

ESI file: HIWINMIKROSYSTEM_ED1F_20250107 or above 

E2 Series EtherCAT Drive 

Software (Thunder): 1.13.7.0 or above 

Firmware: 3.13.6 or above 

ESI file: HIWINMIKROSYSTEM_ED2F_20250107 or above 

ACS SPiiPlusEC 
Software (SPiiPlus MMI Application Studio): 3.13.01 or above 

Firmware: 3.13.01 or above 

1.1 Create new project 

1. Open SPiiPlus MMI Application Studio and click “New workspace” icon. 

 
Figure 1.1.1 

2. Click Add Controller, select Ethernet, and execute Refresh to search for the controller. 
Note: The controller’s default IP address is 10.0.0.100. Correctly set the computer’s IP address to ensure the 

device can be correctly identified. 

 
Figure 1.1.2 

 
Figure 1.1.3 

3. Select the controller and click Connect. Then, the connection will be successfully established. 

 
Figure 1.1.4 

 
Figure 1.1.5 

1.2 Load ESI file 

1. Right-click My Controller and select Add Component→Setup→System Setup. 

 
Figure 1.2.1 

2. Double-click System Setup to get the path where the ESI file should be placed from “Properties” 
column below. 

 
Figure 1.2.2 

3. Open the EtherCAT folder under the Repository Data path and place E series servo drive’s ESI file in 
this path. 

 
Figure 1.2.3 

1.3 Search for the connected device 

1. Double-click System Setup and click Execute to perform the configuration scan of the connected 
device. 

 
Figure 1.3.1  

2. Click Yes to start the configuration scan. The scanning process is shown in the figure below. 

 
Figure 1.3.2  

3. After the configuration scan is completed, ensure that all tasks in “Automatic Setup Task” window are 
checked, then click OK. Once the device is correctly identified, System Configuration will display the 
device name and axis number. 

Note: 

(1) If an error occurs during the configuration scan, re-check the device wiring and execute System Setup again. 

(2) Ensure the controller supports third-party EtherCAT drive by executing #SI in Communication Terminal. 

 
Figure 1.3.3 

 
Figure 1.3.4 

 
 



HIWIN MIKROSYSTEM CORP. 2-1 

2. Parameters Setup 

                                                                                                                                  

 

2. Parameters Setup ............................................................................................................................................ 2-1 

2.1 Select axis and unit ......................................................................................................................... 2-2 

2.2 Set device parameters .................................................................................................................... 2-4 

2.3 Download parameters setup to controller ....................................................................................... 2-9 

 

 
  



Parameters Setup E Series EtherCAT Drive with ACS SPiiPlus MMI Application Studio 

2-2 HIWIN MIKROSYSTEM CORP. 

MD57UE01-2512 

2.1 Select axis and unit 

1. Right-click My Controller and select Add Component→Setup→Adjuster Wizard. 

 
Figure 2.1.1 

2. Double-click Adjuster Wizard to select the axis to be set. Here takes “Axis 0” as an example. 

 
Figure 2.1.2 

3. Click Axis Architecture, configure the structure based on the connected motor type, and select the 
user units. 

 Example of linear motor 

 
Figure 2.1.3 

 Example of rotary motor 

 
Figure 2.1.4 

2.2 Set device parameters 

1. In “Adjuster Wizard” window, go to “Components→Motor” page. Click the edit icon, complete the 
settings based on the motor specification, and click Next >> in the bottom-right corner. 
Here takes “rotary motor DMS03” as an example. 

 
Figure 2.2.1 

2. Go to “Components→Drive” page. Click the edit icon, complete the settings based on the specification 
of drive’s current, and click Next >> in the bottom-right corner. Here takes “E2-3A drive” as an example. 

 
Figure 2.2.2 

3. Go to “Components→Feedback→Motor” page. Click the edit icon, set the resolution based on the 
drive’s control unit, and click Next >> in the bottom-right corner. 
Here takes “DMS03 motor, a resolution of 4325376 cnt/rev” as an example. 

 
Figure 2.2.3 

4. Go to “Components→Calculate Parameters” page to execute Calculate Parameters. After the 
calculation is completed, click Apply Changes to import the device parameters into the settings. 

 
Figure 2.2.4 

 
Figure 2.2.5 

5. Go to “Miscellaneous Definition→Enable / Disable / Brake” page to set Enable Delay time based on 
the motor type. When using the drive’s electrical angle detection method of SW method 1, it is 
recommended to set Enable Delay to 2 seconds or more. 

Note: 

(1) The electrical angle detection method for absolute encoder or Hall sensor is different; the Enable Delay 

time can remain at the default value. 

(2) For linear motor or direct drive motor, the setting of Enable Delay must be adjusted. If not adjusted, an 

error (Error 5019) will be triggered when enabling the motor. 

 
Figure 2.2.6 

6. Go to “Verification→Feedback→Motor” page. Rotate the motor to ensure that the position feedback 
matches the control unit. Here takes “DMS03 motor, a resolution of 4325376 cnt/rev” as an example. 

 
Figure 2.2.7 

2.3 Download parameters setup to controller 

1. In “Adjuster Wizard” window, go to “Save to Flash” page. Check the set axis, click Save to write the 
parameters to the controller, and click Next >> in the bottom-right corner. 

 
Figure 2.3.1 

2. Go to “Final Page” page. Ensure that the Adjuster Wizard setup is completed and click Finish in the 
bottom-right corner. 

 
Figure 2.3.2  

3.1 Motion Manager 

1. Right-click My Controller and select Add Component→Diagnostics and Monitoring→Motion 
Manager. 

 
Figure 3.1.1 

2. Open “Motion Manager” window and select the motion command based on the requirement. 

 
Figure 3.1.2  

3.1.1 Jog Motion 

1. Select the moving axis, set motion parameters, and click Enable. 
Here takes “Axis 0, a velocity of 60 deg/s” as an example. 

 
Figure 3.1.1.1 

2. Click Start Motion. Ensure that the motor performs Jog movement with the velocity of 60 deg/s. 

 
Figure 3.1.1.2  

3.1.2 Relative Move 

1. Click Create Motion and select Relative Move. 

 
Figure 3.1.2.1 

2. Select the moving axis, set motion parameters, and click Enable. 
Here takes “Axis 0, a velocity of 60 deg/s, relative move to the position of 360 deg” as an example. 

 
Figure 3.1.2.2 

3. Click Start Motion. Ensure that the motor performs relative move to the position of 360 deg with the 
velocity of 60 deg/s. 

 
Figure 3.1.2.3 

3.1.3 Absolute Move 

1. Refer to the step of Relative Move. Click Create Motion and select Absolute Move. 

2. Select the moving axis, set motion parameters, and click Enable. 
Here takes “Axis 0, a velocity of 60 deg/s, absolute move to the position of 360 deg” as an example. 

 
Figure 3.1.3.1 

3. Click Start Motion. Ensure that the motor performs absolute move to the position of 360 deg with the 
velocity of 60 deg/s. 

 
Figure 3.1.3.2 

3.2 Homing 

To perform homing, the ACSPL+Programming procedure must be used. Please request the homing 
program file (Single axis homing example.prg) from HIWIN MIKROSYSTEM or copy the following 
program to Program Manager’s Buffer and D-Buffer. 

 Buffer 

!Single-axis HIWIN drive homing process ver1.0 
!!****** Setting Parameter ******!! 
LOCAL INT  X_HmMethod,X_HmType 
LOCAL REAL X_HmVel,X_IndVel,X_HmAcc,X_HmOffset,X_CunitDmt,X_CunitNum,X_UnitType,X_Resolution 
LOCAL INT  ZONE0 
AXISDEF X_axis=0 
!!************************************************************ Operation Unit*****************************************************************************!! 
X_UnitType = 1        ! 0 : Linear(mm). ; 1 : Rotary(deg). 
X_CunitNum = 360       ! Enter the numerator of basic unit for Hiwin drive base on the X_UnitType parameter. 
X_CunitDmt = 4325376       ! Enter the denominator of basic unit for Hiwin drive base on the X_UnitType parameter. 
!!*********************************************** Operation Unit Setup example *******************************************************!! 
!Linear motor with control unit 0.1 um. ***** !Rotary motor with control unit 0.0001 deg. 
!X_UnitType = 0        *****   !X_UnitType = 1 
!X_CunitNum = 1        *****   !X_CunitNum = 360 
!X_CunitDmt = 10000       *****   !X_CunitDmt = 3600000 
!!************************************************************** Basic Settings ***************************************************************************!! 
X_HmMethod = 33        ! Enter the supported homing method. 
X_HmType = 0        ! 0 : HIWIN drive internal homing. ; 1 : Touch probe homing. 
X_HmOffset = 0         ! Homing offset base on the X_UnitType parameter. 
X_HmVel = 30        ! Homing Velocity for finding home sensor base on the X_UnitType parameter. 
X_IndVel = 10         ! Homing Velocity for finding home position base on the X_UnitType parameter. 
X_HmAcc = 300        ! Homing acceleration base on the X_UnitType parameter. 
!!******************************************************************************************************************************************************!! 
!!*************************************************************** Main program **************************************************************************!! 
FCLEAR X_axis 
CALL ECATInOutOffset 
CALL X_DetectDriverError 
MFLAGS(X_axis).#HOME=0 
IF X_HmType=0        ! HIWIN drive internal homing. 
 CALL InterHmProc 
ELSE 
 CALL TouchProbeHmProc      ! Touch probe homing. 
END 
DISP "Disable X_axis error compensation." 
ERRORUNMAP X_axis, ZONE0 
CALL RstMapping 
ENABLE X_axis 
TILL MST(X_axis).#ENABLED 
WAIT 500 
DISP"Axis is moving to home position......" 
PTP/EV X_axis,0,X_HmVel 
MFLAGS(X_axis).#HOME=1 
FDEF(X_axis).#LL = 1 
FDEF(X_axis).#SRL = 1 
FDEF(X_axis).#SLL = 1 
DISP"Axis Homing successful!!" 
STOP 
!!******************************************************************************************************************************************************!! 
!!*************************************************************** InOut mapping *************************************************************************!! 
ECATInOutOffset: 
IF X_HmType = 0 
 ECIN (ECGETOFFSET("Statusword",X_axis),X_StatusWord) 
 ECIN (ECGETOFFSET("Error code",X_axis),X_ErrorCode) 
 ECIN (ECGETOFFSET("Mode of operation display",X_axis),X_ModeOperDisp) 
 ECOUT (ECGETOFFSET("Controlword",X_axis),X_ControlWord) 
 ECOUT (ECGETOFFSET("Mode of operation",X_axis),X_ModeOper) 
ELSE 
 ECOUT (ECGETOFFSET("Touch probe function",X_axis),X_TouchProbFun) 
END 
RET 
STOP 
RstMapping: 
ECUNMAPIN (ECGETOFFSET("Statusword",X_axis)) 
ECUNMAPIN (ECGETOFFSET("Error code",X_axis)) 
ECUNMAPIN (ECGETOFFSET("Mode of operation display",X_axis)) 
ECUNMAPOUT (ECGETOFFSET("Controlword",X_axis)) 
ECUNMAPOUT (ECGETOFFSET("Mode of operation",X_axis)) 
ECUNMAPOUT (ECGETOFFSET("Touch probe function",X_axis)) 
RET 
!!******************************************************************************************************************************************************!! 
!!********************************************************** Drive internal homing process ******************************************************************!! 
InterHmProc: 
DISABLE X_axis 
TILL ^MST(X_axis).#ENABLED 
WAIT 500 
FDEF(X_axis).#LL = 0         ! Motor is not disable when left limit signal rising. 
FDEF(X_axis).#SRL = 0         ! Disable software right limit response. 
FDEF(X_axis).#SLL = 0         ! Disable software left limit response. 
ERRORUNMAP X_axis, -1        ! Close all axis error compensation. 
X_Resolution=1/(X_CunitNum/X_CunitDmt) 

IF X_UnitType = 0         ! Linear type 
 DISP"Linear resolution is %g (cunit/mm)!!", X_Resolution; 
ELSE          ! Rotary type 
 DISP"Rotary resolution is %g (cunit/deg)!!", X_Resolution; 
END 
COEWRITE/1 (X_axis,0x6098,0,X_HmMethod)  
COEWRITE/4 (X_axis,0x6099,1,X_HmVel/(X_CunitNum/X_CunitDmt)) 
COEWRITE/4 (X_axis,0x6099,2,X_IndVel/(X_CunitNum/X_CunitDmt)) 
COEWRITE/4 (X_axis,0x609A,0,X_HmAcc/(X_CunitNum/X_CunitDmt)) 
COEWRITE/4 (X_axis,0x607C,0,X_HmOffset/(X_CunitNum/X_CunitDmt)) 
X_ModeOper = 6         ! Switch to homing mode 
TILL X_ModeOperDisp=6        ! Homing mode switch done 
WAIT 100 
X_ControlWord=6 
TILL X_StatusWord.0=1 
WAIT 600 
X_ControlWord=7 
TILL X_StatusWord.0=1 & X_StatusWord.1=1 
X_ControlWord=15 
TILL X_StatusWord.0=1 & X_StatusWord.1=1 & X_StatusWord.2=1   ! Enabled 
WAIT 100 
X_ControlWord=31        ! Start internal homing (Controlword Bit 4) 
TILL (X_StatusWord.12=1 & X_StatusWord.10=1)|X_StatusWord.13=1   ! Home Finished 
IF X_StatusWord.13=1 
 CALL X_HomeFail 
END 
SET FPOS(X_axis) = COEREAD/4 (X_axis,0x6064,0)*(X_CunitNum/X_CunitDmt) 
X_ControlWord=0 
X_ModeOper=8         ! Switch to CSP mode 
TILL X_ModeOperDisp=8        ! CSP mode switch done 
CALL RstMapping 
WAIT 500 
DISABLE X_axis 
TILL ^MST(X_axis).#ENABLED 
RET 
STOP 
!!******************************************************************************************************************************************************!! 
!!*********************************************************** Touch probe homing process ******************************************************************!! 
!!Search for negative limit switch in reverse direction to find the index 
TouchProbeHmProc: 
DISABLE X_axis 
TILL ^MST(X_axis).#ENABLED 
WAIT 500 
MFLAGS(X_axis).#OPEN = 0       ! Close loop control. 
ERRORUNMAP X_axis, -1        ! Close all axis error compensation. 
MFLAGS(X_axis).#GANTRY = 0       ! Turn off gantry mode (MFLAGS.25). 
FDEF(X_axis).#SRL = 0         ! Disable software right limit response. 
FDEF(X_axis).#SLL = 0         ! Disable software left limit response. 
FDEF(X_axis).#LL = 0         ! Motor is not disable when left limit signal rising. 
ENABLE X_axis 
TILL MST(X_axis).#ENABLED 
WAIT 500 
DISP"Start to find the left limit......" 
JOG/V X_axis,-X_HmVel        ! Find the left limit in negative direction. 
TILL (COEREAD/4 (X_axis,0x60FD,0)& 1) = 1 
HALT X_axis         ! if reach left limit , stop motion 
DISP"Already found the left limit!!" 
WAIT 100 
DISP"Start to leave the left limit......" 
JOG/V X_axis,X_IndVel 
TILL (COEREAD/4 (X_axis,0x60FD,0)& 1) = 0 
DISP"Already leave the left limit!!" 
X_TouchProbFun = 0        ! Reset touch probe funtion 
WAIT 200 
X_TouchProbFun = 21        ! Start touch probe function 
WAIT 200 
DISP"Start to find the index in positive direction......"     ! Find the index in positive direction 
TILL (COEREAD/2 (X_axis,0x60B9,0) = 3)      ! Stop motion when index position is reached. 
HALT X_axis 
TILL ^MST(X_axis).#MOVE 
DISP"Already found the index!!" 
SET FPOS(X_axis) = (COEREAD/4 (X_axis,0x6064,0) - COEREAD/4 (X_axis,0x60BA,0))*(X_CunitNum/X_CunitDmt) + X_HmOffset  !Set index position as zero. 
RET 
STOP 
!!******************************************************************************************************************************************************!! 
!!***************************************************************** Error detection************************************************************************!! 
X_DetectDriverError: 
IF X_ErrorCode<>0 
 DISP "Drive has error %x , Please check the driver",X_ErrorCode 
 CALL X_HomeFail 
END 
RET 
STOP 
X_HomeFail: 
DISP"Homing process fail. " 
HALT X_axis 
X_ControlWord=0 
WAIT 100 
X_ModeOper = 8 
TILL X_ModeOperDisp=8 
WAIT 100 
CALL RstMapping 
STOP 

 D-Buffer 

GLOBAL INT X_TouchProbFun,X_StatusWord,X_ControlWord,X_ModeOper,X_ModeOperDisp,X_ErrorCode 
global int I(100),I0,I1,I2,I3,I4,I5,I6,I7,I8,I9,I90,I91,I92,I93,I94,I95,I96,I97,I98,I99 
global real V(100),V0,V1,V2,V3,V4,V5,V6,V7,V8,V9,V90,V91,V92,V93,V94,V95,V96,V97,V98,V99 

3.2.1 Program Manager 

1. Right-click My Controller and select Add Component→Application Development→Program 
Manager. 

 
Figure 3.2.1.1 

2. Click Program Manager’s “Load from File” icon, select the homing program file, and click Load. 

 
Figure 3.2.1.2 

Figure 3.2.1.3 

3. Click OK to load the homing program file. 

 
Figure 3.2.1.4 

3.2.2 Communication Terminal 

1. Right-click My Controller and select Add Component→Utilities→Communication Terminal. 

 
Figure 3.2.2.1 

2. After successfully opening Communication Terminal, set Show Messages to On. 

 
Figure 3.2.2.2  

3.2.3 Execute homing 

1. Click Program Manager’s “Buffer” icon to complete homing’s parameters setup and execute 
compilation. Here takes “DMS03 motor, a resolution of 4325376 cnt/rev, drive’s internal homing 
method 33” as an example. 

Note: 

(1) Correctly set X_UnitType, X_CunitNum, X_CunitDmt, X_HmMethod, and X_HmType; otherwise, homing 

conditions may differ from the expected motion conditions. 

(2) “mm” is for linear unit; “deg” is for rotary unit. Set motion conditions with this fixed unit. 

(3) Set X_CunitNum and X_CunitDmt based on the drive’s control unit. Refer to the homing program for the 

setting example. 

(4) X_HmType = 0 indicates the drive’s internal homing; X_HmType = 1 indicates Touch Probe homing. 

(5) Do not key in the method unsupported by the drive’s internal homing to X_HmMethod. The example of 

Touch Probe homing is fixed: search for Index position in the reverse direction after touching the negative 

limit. 

 
Figure 3.2.3.1 

2. Click “Run” icon to start homing. 

 
Figure 3.2.3.2 

3. Homing starts. Use Communication Terminal to ensure that homing is successfully executed and that 
the motor has indeed returned to the home position. 

 
Figure 3.2.3.3 

4.1 Read object parameters 

EtherCAT communication objects can be read via COEREAD command in ACSPL+Commands. 
Refer to ACS’s Knowledge Center for detailed command description. 

1. Open Program Manager’s Buffer and key in COEREAD command to read the object value. 
Here takes “reading drive parameter Pt100” as an example. 

DISP "Pt100=%d",COEREAD/2 (0,0x2100,0); 
STOP 

 

 
Figure 4.1.1 

2. Compile it and click “Run” icon. The object value will be displayed in Communication Terminal. 

 
Figure 4.1.2 

4.2 Write object parameters 

EtherCAT communication objects can be written via COEWRITE command in ACSPL+Commands. 
Refer to ACS’s Knowledge Center for detailed command description. 

1. Open Program Manager’s Buffer and key in COEWRITE command to modify the object value. 
Here takes “modifying drive parameter Pt100” as an example. 

COEWRITE/2 (0,0x2100,0,1000) 
STOP 

 

 
Figure 4.2.1 

2. Compile it and click “Run” icon. Ensure that the value of parameter Pt100 has been correctly 
modified. The command of reading object parameters can be used for verification. 

 
Figure 4.2.2 

